Pensieve header: Computations in the ybax algebra using the Drinfel'd double (at $\epsilon^2=0$. Continues ExpDoubleEpsilonSquare5@@.nb.

The double at $\epsilon^2 = 0$

Utilities

Canonical Form:

```
In[•]:=
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
```

The Kronecker δ:

```
In[•]:=
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality and multiplication of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

```
In[•]:=
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

Zip and Bind

```
In[•]:=
{b*, y*, a*, x*, z*} = {β, η, α, ξ, ζ};
{β*, η*, α*, ξ*, ζ*} = {b, y, a, x, z}; (u_i_)* := (u*)_i;
```

```
In[•]:=
expand[sd_SeriesData] := MapAt[expand, sd, 3];
expand[ℰ_] := Expand[ℰ];
Zip_{}[P_] := P;
Zip_{ζ_,ζs___}[P_] := (expand[P // Zip_{ζs}] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

```
In[•]:=
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zip_ζs[e^Q1 (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

Upper to lower and lower to Upper:

In[•]:=
```
U2l = {B^p_._{i_} → e^(-p b_i), B^p_. → e^(-p b), 𝒜^p_._{i_} → e^(p α_i), 𝒜^p_. → e^(p α)};
l2U = {e^(c_. b_{i_}+d_.) :→ B_i^(-c) e^d, e^(c_. b+d_.) :→ B^(-c) e^d,
    e^(c_. α_{i_}+d_.) :→ 𝒜_i^c e^d, e^(c_. α+d_.) :→ 𝒜^c e^d,
    e^(ℰ_) :→ e^(Expand@ℰ)};
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single"P". Here the z's are b and α and the ζ's are β and a.

In[•]:=
```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    simp /@ 𝔼[L2, Q2, Det[lt] e^(-L2-Q2) Zip_ζs[e^(L1+Q1) (P /. U2l /. zrule)]] //. l2U];
LZip_{ζs_List} := LZip_{ζs,CF};
```

In[•]:=
```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
    Times[
        L /. Table[(v : b | B | a | x | y)_i → v_{n@i}, {i, {is}}],
        R /. Table[(v : β | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Flatten@Table[{β_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{l_List}[L_, R_] := Bind_l[L, R]; B_{is___}[L_, R_] := Bind_{is}[L, R];
```

The two halves

In[∘]:=

```
(*Hopf algebra on the a,x side*)
am_{i_,j_->k_} := 𝔼[(α_i + α_j) a_k, (ⅇ^{-α_j} ξ_i + ξ_j) x_k, 1 + O[ϵ]^2]
aΔ_{i_->j_,k_} := 𝔼[α_i (a_j + a_k), ξ_i (x_j + x_k), 1 + ϵ ξ_i x_k (-a_j + 1/2 ξ_i x_j) + O[ϵ]^2]
aS_{i_} := 𝔼[-α_i a_i, -ⅇ^{α_i} ξ_i x_i, 1 - ϵ ⅇ^{α_i} ξ_i x_i (a_i + 1/2 ⅇ^{α_i} ξ_i x_i) + O[ϵ]^2]
aSi_{i_} := 𝔼[-α_i a_i, -ⅇ^{α_i} ξ_i x_i, 1 - ϵ ⅇ^{α_i} ξ_i x_i (a_i - 1 + 1/2 ⅇ^{α_i} ξ_i x_i) + O[ϵ]^2]
(*Hopf algebra on the y,b side*)
bm_{i_,j_→k_} := 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, 1 - ϵ η_j y_k β_i + O[ϵ]^2]
bΔ_{i_->j_,k_} := 𝔼[β_i (b_j + b_k), η_i (ⅇ^{-b_k} y_j + y_k), 1 + 1/2 ϵ η_i^2 y_j y_k ⅇ^{-b_k} + O[ϵ]^2]
bS_{i_} := 𝔼[-β_i b_i, -ⅇ^{b_i} η_i y_i, 1 - ϵ ⅇ^{b_i} η_i y_i (β_i + 1/2 ⅇ^{b_i} η_i y_i) + O[ϵ]^2]
bSi_{i_} := 𝔼[-β_i b_i, -ⅇ^{b_i} η_i y_i, 1 - ϵ ⅇ^{b_i} η_i y_i (β_i - 1 + 1/2 ⅇ^{b_i} η_i y_i) + O[ϵ]^2]
```

First check that on the generators this agrees with our conventions in SLPortfolio.pdf with $\hbar = \gamma = 1$

In[∘]:=

```
{
  "[a,x]" → ((𝔼[0, 0, a_2 x_1] ~B_{1,2}~ am_{1,2→1})⟦3⟧ - (𝔼[0, 0, a_1 x_2] ~B_{1,2}~ am_{1,2→1})⟦3⟧),
  "[b,y]" → ((𝔼[0, 0, y_2 b_1] ~B_{1,2}~ bm_{1,2→1})⟦3⟧ - (𝔼[0, 0, y_1 b_2] ~B_{1,2}~ bm_{1,2→1})⟦3⟧)
} /. {z_1 → z} // Simplify
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #_1] ~B_1~ aΔ_{1→1,2}]) & /@ {a, x}
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #_1] ~B_1~ bΔ_{1→1,2}]) & /@ {b, y}
{
  " S(a) = " ((𝔼[0, 0, a_1] ~B_1~ aS_1)⟦3⟧),
  " S(x) = " ((𝔼[0, 0, x_1] ~B_1~ aS_1)⟦3⟧),
  " S(b) = " ((𝔼[0, 0, b_1] ~B_1~ bS_1)⟦3⟧),
  " S(y) = " → ((𝔼[0, 0, y_1] ~B_1~ bS_1)⟦3⟧)
} /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,x] \to -x + O[\epsilon]^2,\ [b,y] \to -y\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta[a] \to a_1 + a_2,\ \Delta[x] \to x_1 + (1 - \epsilon\, a_1)\, x_2\}$

Out[∘]= $\{\Delta[b] \to b_1 + b_2,\ \Delta[y] \to B_2\, y_1 + y_2\}$

Out[∘]= $\{-\ S(a)\ =\ a + O[\epsilon]^2,\ -\ S(x)\ =\ x -\ S(x)\ =\ a\,x\,\epsilon + O[\epsilon]^2,$
$-\ S(b)\ =\ b + O[\epsilon]^2,\ \ S(y)\ =\ \to -\frac{y}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

```
In[•]:= {(aΔ1→1,2 ~ B2 ~ aΔ2→2,3) ≡ (aΔ1→1,3 ~ B1 ~ aΔ1→1,2), (bΔ1→1,2 ~ B2 ~ bΔ2→2,3) ≡ (bΔ1→1,3 ~ B1 ~ bΔ1→1,2),
  (am1,2→1 ~ B1 ~ am1,3→1) ≡ (am2,3→2 ~ B2 ~ am1,2→1), (bm1,2→1 ~ B1 ~ bm1,3→1) ≡ (bm2,3→2 ~ B2 ~ bm1,2→1)}
Out[•]= {True, True, True, True}
```

Δ is an algebra morphism

```
In[•]:= {am1,2→1 ~ B1 ~ aΔ1→1,2 ≡ (aΔ1→1,3 aΔ2→2,4) ~ B1,2,3,4 ~ (am3,4→2 am1,2→1),
  bm1,2→1 ~ B1 ~ bΔ1→1,2 ≡ (bΔ1→1,3 bΔ2→2,4) ~ B1,2,3,4 ~ (bm3,4→2 bm1,2→1)}
Out[•]= {True, True}
```

S is convolution inverse of id

```
In[•]:= {(aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am1,2→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am1,2→1}
  {(bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm1,2→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm1,2→1}
Out[•]= {𝔼[0, 0, 1 + O[ϵ]^2], 𝔼[0, 0, 1 + O[ϵ]^2]}
Out[•]= {𝔼[0, 0, 1 + O[ϵ]^2], 𝔼[0, 0, 1 + O[ϵ]^2]}
```

Si is the inverse of S

```
In[•]:= {aSi1 ~ B1 ~ aS1 ≡ 𝔼[a1 α1, x1 ξ1, 1], aS1 ~ B1 ~ aSi1 ≡ 𝔼[a1 α1, x1 ξ1, 1]}
  {bSi1 ~ B1 ~ bS1 ≡ 𝔼[b1 β1, y1 η1, 1], bS1 ~ B1 ~ bSi1 ≡ 𝔼[b1 β1, y1 η1, 1]}
Out[•]= {True, True}
Out[•]= {True, True}
```

S is an algebra anti-(co)morphism

```
In[•]:= {am1,2→1 ~ B1 ~ aS1 ≡ (aS1 aS2) ~ B1,2 ~ am2,1→1, bm1,2→1 ~ B1 ~ bS1 ≡ (bS1 bS2) ~ B1,2 ~ bm2,1→1}
  {aS1 ~ B1 ~ aΔ1→1,2 ≡ aΔ1→2,1 ~ B1,2 ~ (aS1 aS2), bS1 ~ B1 ~ bΔ1→1,2 ≡ bΔ1→2,1 ~ B1,2 ~ (bS1 bS2)}
Out[•]= {True, True}
Out[•]= {True, True}
```

Pairing

```
In[•]:= tP_{i_,j_} := 𝔼[β_i α_j, η_i ξ_j, 1 + 1/4 ϵ η_i^2 ξ_j^2]
```

```
In[•]:= qfac[k_, q_] := (1 - q)^-k QPochhammer[q, q, k] // FunctionExpand
  qfe[k_] := Normal[Series[qfac[k, E^ρ], {ρ, 0, 1}]] /. {ρ → ϵ}
  Table[𝔼[0, 0, y1^r b1^s a2^t x2^u] ~ B1,2 ~ tP1,2 ≡ 𝔼[0, 0, Kδ_{r,u} Kδ_{s,t} qfe[r] s!],
    {r, 0, 4}, {s, 0, 4}, {t, 0, 4}, {u, 0, 4}] // Flatten // Union
Out[•]= {True}
```

Pairing axioms

In[]:=
```
{(bm1,2→1 𝔼[α3 a3, ξ3 x3, 1]) ~ B1,3 ~ tP1,3 ≡
   (𝔼[β1 b1, η1 y1, 1] 𝔼[β2 b2, η2 y2, 1] aΔ3→4,5) ~ B1,4 ~ tP1,4 ~ B2,5 ~ tP2,5
 , (bΔ1→1,2 𝔼[α3 a3, ξ3 x3, 1] 𝔼[α4 a4, ξ4 x4, 1]) ~ B1,3 ~ tP1,3 ~ B2,4 ~ tP2,4 ≡
   (𝔼[β1 b1, η1 y1, 1] am3,4→3) ~ B1,3 ~ tP1,3 }
```

Out[]= {True, True}

In[]:=
```
{(bS1 𝔼[α2 a2, ξ2 x2, 1]) ~ B1,2 ~ tP1,2 ≡ (𝔼[β1 b1, η1 y1, 1] aS2) ~ B1,2 ~ tP1,2,
 (bSi1 𝔼[α2 a2, ξ2 x2, 1]) ~ B1,2 ~ tP1,2 ≡ (𝔼[β1 b1, η1 y1, 1] aSi2) ~ B1,2 ~ tP1,2}
```

Out[]= {True, True}

The Double

The double multiplication (should really bind the a’s and b’s separately)

In[]:=
```
Block[{i, j, k},
 dmi_,j_→k_ = Simplify /@ Expand /@ ((𝔼[βi bi + αj aj, ηi yi + ξj xj, 1] (aΔi→h1,h2 ~ Bh2 ~ aΔh2→h2,h3)
        (bΔj→t1,t2 ~ Bt2 ~ bΔt2→t2,t3)) ~ Bh3 ~ aSih3 ~ Bt1,h3 ~ (tPt1,h3) ~
      Bt3,h1 ~ (tPt3,h1) ~ Bh2,j,i,t2 ~ (amh2,j→k bmi,t2→-k) /. {u_-k :> uk}) ]
```

Out[]=
$$\mathbb{E}\Big[a_k\,(\alpha_i+\alpha_j)+b_k\,(\beta_i+\beta_j),\; y_k\left(\eta_i+\frac{\eta_j}{\mathcal{A}_i}\right)-\left(-1+\mathrm{B}_k\right)\eta_j\,\xi_i+x_k\left(\frac{\xi_i}{\mathcal{A}_j}+\xi_j\right),$$
$$1+\frac{1}{4\,\mathcal{A}_i\,\mathcal{A}_j}\Big(2\,y_k\,\eta_j\,\big(2\,x_k\,\xi_i+\mathcal{A}_j\,\big(-2\,\beta_i+\left(1-3\,\mathrm{B}_k\right)\eta_j\,\xi_i\big)\big)+$$
$$\mathcal{A}_i\,\xi_i\,\big(x_k\,\big(-4\,\beta_j+2\,\left(1-3\,\mathrm{B}_k\right)\eta_j\,\xi_i\big)+\mathcal{A}_j\,\eta_j\,\big(4\,a_k\,\mathrm{B}_k+\left(1-4\,\mathrm{B}_k+3\,\mathrm{B}_k^2\right)\eta_j\,\xi_i\big)\big)\Big)\,\epsilon+O[\epsilon]^2\Big]$$

In[]:=
```
(*Deriving dS using dm*)
Block[{i}, dSi_ = ((bSi1 aS2) ~ B1,2 ~ dm2,1→i) /. {z_1|2 → zi}]
```

Out[]=
$$\mathbb{E}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\;\frac{-y_i\,\mathcal{A}_i\,\eta_i-\mathrm{B}_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-\mathrm{B}_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\mathrm{B}_i},$$
$$1+\frac{1}{4\,\mathrm{B}_i^2}\Big(4\,\mathrm{B}_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,\mathrm{B}_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,a_i\,\mathrm{B}_i^2\,x_i\,\mathcal{A}_i\,\xi_i-$$
$$4\,\mathrm{B}_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-4\,\mathrm{B}_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,a_i\,\mathrm{B}_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\mathrm{B}_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,\mathrm{B}_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+$$
$$4\,\mathrm{B}_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,\mathrm{B}_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\mathrm{B}_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\mathrm{B}_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+$$
$$6\,\mathrm{B}_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,\mathrm{B}_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,\mathrm{B}_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-\mathrm{B}_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

In[]:=
```
(*Deriving dΔ using dm*)
Block[{i, j, k}, dΔi_→j_,k_ = (bΔi→3,1 aΔi→2,4) ~ B1,2,3,4 ~ (dm3,4→k dm1,2→j)]
```

Out[]=
$$\mathbb{E}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\;y_j\,\eta_i+\mathrm{B}_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,$$
$$1+\frac{1}{2}\left(\mathrm{B}_j\,y_j\,y_k\,\eta_i^2-2\,a_j\,x_k\,\xi_i+x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[∘]:= {
      "[a,y]" → ((𝔼[0, 0, y₂ a₁] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ a₂] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧),
      "[b,x]"  → ((𝔼[0, 0, x₂ b₁] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, x₁ b₂] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧),
      "xy-qyx" → ((𝔼[0, 0, x₁ y₂] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧ - (1 + ϵ) (𝔼[0, 0, y₁ x₂] ~ B₁,₂ ~ dm₁,₂→₁)⟦3⟧)
     } /. {z_1 → z} // Expand // Factor
    {
     "Δ(a)" → ((𝔼[0, 0, a₁] ~ B₁ ~ dΔ₁→₁,₂)⟦3⟧),
     "Δ(x)" → ((𝔼[0, 0, x₁] ~ B₁ ~ dΔ₁→₁,₂)⟦3⟧),
     "Δ(b)" → ((𝔼[0, 0, b₁] ~ B₁ ~ dΔ₁→₁,₂)⟦3⟧),
     "Δ(y)" → ((𝔼[0, 0, y₁] ~ B₁ ~ dΔ₁→₁,₂)⟦3⟧)
    } // Simplify
    {
      "S(a)" → ((𝔼[0, 0, a₁] ~ B₁ ~ dS₁)⟦3⟧),
      "S(x)" → ((𝔼[0, 0, x₁] ~ B₁ ~ dS₁)⟦3⟧),
      "S(b)" → ((𝔼[0, 0, b₁] ~ B₁ ~ dS₁)⟦3⟧),
      "S(y)" → ((𝔼[0, 0, y₁] ~ B₁ ~ dS₁)⟦3⟧)
     } /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,y] \to -y + O[\epsilon]^2,\ [b,x] \to x\,\epsilon + O[\epsilon]^2,\ xy\text{-}qyx \to (1 - B) + a\,B\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^2,\ \Delta(x) \to (x_1 + x_2) - a_1\,x_2\,\epsilon + O[\epsilon]^2,$
$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^2,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^2\}$

Out[∘]= $\{S(a) \to -a + O[\epsilon]^2,\ S(x) \to -x - a\,x\,\epsilon + O[\epsilon]^2,\ S(b) \to -b + O[\epsilon]^2,\ S(y) \to -\frac{y}{B} + \frac{y\,\epsilon}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on double

(co)-associativity

```
In[∘]:= {(dΔ₁→₁,₂ ~ B₂ ~ dΔ₂→₂,₃) ≡ (dΔ₁→₁,₃ ~ B₁ ~ dΔ₁→₁,₂),
     (dm₁,₂→₁ ~ B₁ ~ dm₁,₃→₁) ≡ (dm₂,₃→₂ ~ B₂ ~ dm₁,₂->₁)}
```

Out[∘]= {True, True}

Δ is an algebra morphism

```
In[∘]:= dm₁,₂→₁ ~ B₁ ~ dΔ₁→₁,₂ ≡ (dΔ₁→₁,₃ dΔ₂→₂,₄) ~ B₁,₂,₃,₄ ~ (dm₃,₄→₂ dm₁,₂→₁)
```

Out[∘]= True

S is convolution inverse of id

```
In[∘]:= {(dΔ₁→₁,₂ ~ B₁ ~ dS₁) ~ B₁,₂ ~ dm₁,₂→₁, (dΔ₁→₁,₂ ~ B₂ ~ dS₂) ~ B₁,₂ ~ dm₁,₂→₁}
```

Out[∘]= $\{\mathbb{E}[0,\ 0,\ 1 + O[\epsilon]^2],\ \mathbb{E}[0,\ 0,\ 1 + O[\epsilon]^2]\}$

S is a (co)-algebra anti-morphism

```
In[∘]:= {dm₁,₂→₁ ~ B₁ ~ dS₁ ≡ (dS₁ dS₂) ~ B₁,₂ ~ dm₂,₁→₁, dS₁ ~ B₁ ~ dΔ₁→₁,₂ ≡ dΔ₁→₂,₁ ~ B₁,₂ ~ (dS₁ dS₂)} // Expand
```

Out[∘]= {True, True}

R-matrix

In[∘]:=
```
𝕖_{q_,k_}[x_] := 𝕖^(Sum_{j=1}^{k+1} ((1-q)^j x^j)/(j (1-q^j)))
R_{i_,j_} := 𝔼[b_i a_j, y_i x_j, 1 - ϵ 1/4 y_i^2 x_j^2 + O[ϵ]^2]
  (*First two terms in Faddeev-Quesne formula*)
```

In[∘]:=
```
Series[𝕖_{q,1}[z] /. {z → y_i x_j, q → 1 + ρ}, {ρ, 0, 1}] /. {ρ → ϵ}
```

Out[∘]= $e^{x_j y_i} - \frac{1}{4}\left(e^{x_j y_i} x_j^2 y_i^2\right) \epsilon + O[\epsilon]^2$

Quasi-triangular axiom 1:

In[∘]:=
```
R_{1,2} ~B_1~ dΔ_{1->1,3} ≡ (R_{1,4} R_{3,2}) ~B_{2,4}~ dm_{2,4→2}
```

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:=
```
((dΔ_{1->1,2} R_{3,4}) ~B_{1,2,3,4}~ (dm_{1,3→1} dm_{2,4→2})) ≡
 ((dΔ_{1→2,1} R_{3,4}) ~B_{1,2,3,4}~ (dm_{3,1→1} dm_{4,2→2}))
```

Out[∘]= True

Reidemeister 3:

In[∘]:=
```
((R_{1,2} R_{4,3} R_{5,6}) ~B_{1,4}~ dm_{1,4→1} ~B_{2,5}~ dm_{2,5→2} ~B_{3,6}~ dm_{3,6→3}) ≡
 ((R_{1,6} R_{2,3} R_{4,5}) ~B_{1,4}~ dm_{1,4→1} ~B_{2,5}~ dm_{2,5→2} ~B_{3,6}~ dm_{3,6→3})
```

Out[∘]= True

In[∘]:=
```
Block[{i, j}, R̄_{i_,j_} = Expand /@ R_{i,j} ~B_j~ dS_j]
```

Out[∘]= $\mathbb{E}\left[-a_j b_i, -\frac{x_j y_i}{B_i}, 1 + \frac{\left(-4 a_j B_i x_j y_i - 3 x_j^2 y_i^2\right) \epsilon}{4 B_i^2} + O[\epsilon]^2\right]$

Reidemeister 2

In[∘]:=
```
{(R̄_{1,2} R_{3,4}) ~B_{1,2,3,4}~ (dm_{1,3→1} dm_{2,4→2}), (R_{1,2} R̄_{3,4}) ~B_{1,2,3,4}~ (dm_{1,3→1} dm_{2,4→2})}
```

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2], \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

Deriving the Drinfeld element u and its inverse ui

In[∘]:=
```
Block[{i}, {
   u_{i_} = R_{1,2} ~B_1~ dS_1 ~B_{1,2}~ dm_{2,1→i},
   ui_{i_} := R_{1,2} ~B_2~ dS_2 ~B_2~ dS_2 ~B_{1,2}~ dm_{2,1→i}
  }]
```

Out[∘]= $\left\{\mathbb{E}\left[-a_i b_i, -\frac{x_i y_i}{B_i}, B_i + \frac{\left(-4 a_i B_i^2 - 4 B_i x_i y_i - 4 a_i B_i x_i y_i - 3 x_i^2 y_i^2\right) \epsilon}{4 B_i} + O[\epsilon]^2\right], \text{Null}\right\}$

u and ui are inverses

In[∘]:= $(u_1\, ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0, 0, 1 + O[\epsilon]^2\right]$

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner $C=uv^{-1}$.
It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= $((u_1 \sim B_1 \sim dS_1)\; ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0, 0, \frac{1}{B_1} + \frac{a_1\,\epsilon}{B_1} + O[\epsilon]^2\right]$

(* Needs fixing! *) So in our case S(u) = u z so S(u)u = $u^2 z$ and v = $uz^{\frac{1}{2}}$ and finally C = uv^{-1} =
$z^{-\frac{1}{2}} = e^{t_1/2}(1 - \epsilon\, a_1)$.

In[∘]:=
$$\begin{aligned}&\text{Block}[\{i\},\\ &\{CC_{i\_} = \mathbb{E}\left[0, 0, B_i^{1/2}\, e^{-\epsilon\, a_i/2} + O[\epsilon]^2\right],\\ &\overline{CC}_{i\_} = \mathbb{E}\left[0, 0, B_i^{-1/2}\, e^{\epsilon\, a_i/2} + O[\epsilon]^2\right]\\ &\}]\end{aligned}$$

Out[∘]= $\left\{\mathbb{E}\left[0, 0, \sqrt{B_i} - \frac{1}{2}\left(a_i\sqrt{B_i}\right)\epsilon + O[\epsilon]^2\right], \mathbb{E}\left[0, 0, \frac{1}{\sqrt{B_i}} + \frac{a_i\,\epsilon}{2\sqrt{B_i}} + O[\epsilon]^2\right]\right\}$

In[∘]:=
$$\begin{aligned}&\text{Block}[\{i, j\}, \{\\ &\text{Kink}_{i\_} = (R_{1,3}\, \overline{CC}_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i},\\ &\overline{\text{Kink}}_{j\_} = (\overline{R}_{1,3}\, CC_2) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}\\ &\}]\end{aligned}$$

Out[∘]=
$$\left\{\mathbb{E}\left[a_i\, b_i, x_i\, y_i, \frac{1}{\sqrt{B_i}} + \frac{\left(2\, a_i - x_i^2\, y_i^2\right)\epsilon}{4\sqrt{B_i}} + O[\epsilon]^2\right],\right.$$
$$\left.\mathbb{E}\left[-a_j\, b_j, -\frac{x_j\, y_j}{B_j}, \sqrt{B_j} + \frac{\left(-2\, a_j\, B_j^2 - 4\, a_j\, B_j\, x_j\, y_j - 3\, x_j^2\, y_j^2\right)\epsilon}{4\, B_j^{3/2}} + O[\epsilon]^2\right]\right\}$$

In[∘]:=
```
k2 = (R3,1 CC2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→i /. e → E;
k4 = (R̄3,1 C̄C̄2) ~ B1,2 ~ dm1,2→1 ~ B1,3 ~ dm1,3→j /. e → E;
Simplify@{Kinki ≡ k2, K̄inkj ≡ k4, (Kinki K̄inkj) ~ Bi,j ~ dmi,j→1}
```

Out[∘]= $\{\text{True}, \text{True}, \mathbb{E}\left[0, 0, 1 + O[\epsilon]^2\right]\}$

Reidemeister 2:

In[∘]:= $(R_{1,2}\, \overline{R}_{3,4}) \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{2,4} \sim dm_{2,4\to 2}$

Out[∘]= $\mathbb{E}\left[0, 0, 1 + O[\epsilon]^2\right]$

Cyclic Reidemeister 2:

In[◦]:= $\left(R_{1,4}\,\overline{R}_{5,2}\,\overline{CC}_3\right) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \overline{CC}_1$

Out[◦]= True

Trefoil

In[◦]:= Timing[
 $Z = R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10}$;
 Do[Print["doing ", r]; $Z = Z \sim B_{1,r} \sim dm_{1,r\to 1}$, {r, 2, 10}];
 Simplify@Z[[3]]]

```
doing 2
doing 3
doing 4
doing 5
doing 6
doing 7
doing 8
doing 9
doing 10
```

Out[◦]= $\left\{70.7344, \frac{B_1}{1 - B_1 + B_1^2} + \left(B_1 \left(-B_1 + 2 B_1^2 + 2 B_1^4 + a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) - 2 x_1 y_1 - B_1^3 \left(3 + 2 x_1 y_1\right)\right) \in\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\in]^2\right\}$